In[1]:= **(* Define a patch. *)**

x = {(v^2 + 1) Cos[u], (v^2 + 1) Sin[u], v}

(* Clearly x is smooth, and one-to-one on the strip where $-\pi<u<\pi$. *)

Out[1]= $\{(1+v^2)\,\text{Cos}[u],\ (1+v^2)\,\text{Sin}[u],\ v\}$

In[2]:= **(* Compute partial velocities.*)**

x_u = D[x, u]

x_v = D[x, v]

Out[2]= $\{-(1+v^2)\,\text{Sin}[u],\ (1+v^2)\,\text{Cos}[u],\ 0\}$

Out[3]= $\{2\,v\,\text{Cos}[u],\ 2\,v\,\text{Sin}[u],\ 1\}$

In[4]:= **(* Check for regularity. *)**

$x_u \times x_v$ // Simplify

Out[4]= $\{(1+v^2)\,\text{Cos}[u],\ (1+v^2)\,\text{Sin}[u],\ -2\,(v+v^3)\}$

(* Since $1+v^2$ is never zero, and cos(u) and sin(u) can't be simultaneously zero, $x_u \times x_v$ is clearly never zero, and thus x is regular. *)

In[5]:= **(* Compute E,F, and G for patch x. *)**

eE = $x_u . x_u$

fF = $x_u . x_v$

gG = $x_v . x_v$

Out[5]= $(1+v^2)^2\,\text{Cos}[u]^2 + (1+v^2)^2\,\text{Sin}[u]^2$

Out[6]= 0

Out[7]= $1 + 4\,v^2\,\text{Cos}[u]^2 + 4\,v^2\,\text{Sin}[u]^2$

In[8]:= **(* Express the 1st fundamental form in local coordinates. *)**

g = {{eE, fF}, {fF, gG}} // Simplify;

g // MatrixForm

Out[9]//MatrixForm= $\begin{pmatrix} (1+v^2)^2 & 0 \\ 0 & 1+4\,v^2 \end{pmatrix}$

In[10]:= **(* Label the determinant...since it shows up in more than one formula. *)**

w = Det[g]

Out[10]= $1 + 6\,v^2 + 9\,v^4 + 4\,v^6$

In[11]:= **(* Compute unit normal in local coordinates. *)**

uU = $\dfrac{x_u \times x_v}{\sqrt{w}}$ // FullSimplify[#, Assumptions → {u ∈ Reals, v ∈ Reals}] &

Out[11]= $\left\{\dfrac{\text{Cos}[u]}{\sqrt{1+4\,v^2}},\ \dfrac{\text{Sin}[u]}{\sqrt{1+4\,v^2}},\ -\dfrac{2\,v}{\sqrt{1+4\,v^2}}\right\}$

(* This is equivalent to $\dfrac{x_u \times x_v}{\text{Norm}[x_u \times x_v]}$.)

In[12]:=
```
(* Compute the "partials of the partial velocities." *)
```
$x_{uu} = D[x_u, u]$

$x_{uv} = D[x_u, v]$

$x_{vv} = D[x_v, v]$

Out[12]= $\{-(1+v^2)\,Cos[u],\ -(1+v^2)\,Sin[u],\ 0\}$

Out[13]= $\{-2\,v\,Sin[u],\ 2\,v\,Cos[u],\ 0\}$

Out[14]= $\{2\,Cos[u],\ 2\,Sin[u],\ 0\}$

In[15]:=
```
(* And L,M, and N. *)
```
$lL = x_{uu}.uU$

$mM = x_{uv}.uU$

$nN = x_{vv}.uU$

Out[15]= $-\frac{(1+v^2)\,Cos[u]^2}{\sqrt{1+4\,v^2}} - \frac{(1+v^2)\,Sin[u]^2}{\sqrt{1+4\,v^2}}$

Out[16]= 0

Out[17]= $\frac{2\,Cos[u]^2}{\sqrt{1+4\,v^2}} + \frac{2\,Sin[u]^2}{\sqrt{1+4\,v^2}}$

In[18]:=
```
(* So the 2nd fundamental form, in local coordinates, is... *)
iII = {{lL, mM}, {mM, nN}} // Simplify;
iII // MatrixForm
```

Out[19]//MatrixForm= $\begin{pmatrix} -\frac{1+v^2}{\sqrt{1+4\,v^2}} & 0 \\ 0 & \frac{2}{\sqrt{1+4\,v^2}} \end{pmatrix}$

In[20]:=
```
(* So Gaussian and mean curvature [in terms of our local coordinates] are ... *)
```
$kK = \frac{Det[iII]}{w}$ // FullSimplify

$hH = \frac{gG\ lL + eE\ nN - 2\ fF\ mM}{2\,w}$ // FullSimplify

Out[20]= $-\frac{2}{(1+v^2)\,(1+4\,v^2)^2}$

Out[21]= $\frac{1-2\,v^2}{2\,(1+v^2)\,(1+4\,v^2)^{3/2}}$

(* Notice that K is strictly negative, has maximum magnitude when v=0, and goes to 0 as v→∞. *)

(* H switches sign upon crossing the $v=\frac{1}{\sqrt{2}}$ and $v=\frac{-1}{\sqrt{2}}$ parameter curves. *)

(* Do these observations appear to be consistent with the plot below. *)

In[22]:=
```
ParametricPlot3D[x, {u, -π, π}, {v, -1, 1}]
```

Out[22]=